
Shashank Tiwari
Elad Elrom

Novatec

Flex 4
Avançado



Original English language edition published by Friends of ED, an imprint of Apress Inc., Copyright © 2010 by 
Apress, Inc.. Portuguese-language edition for Brazil copyright © 2011 by Novatec Editora. All rights reserved.

Edição original em Inglês publicada pela Friends of ED, uma marca da Apress Inc., Copyright © 2010 pela Apress, 
Inc.. Edição em Português para o Brasil copyright © 2011 pela Novatec Editora. Todos os direitos reservados.

© Novatec Editora Ltda. 2011.

Todos os direitos reservados e protegidos pela Lei 9.610 de 19/02/1998. 
É proibida a reprodução desta obra, mesmo parcial, por qualquer processo, sem prévia autorização, por 
escrito, do autor e da Editora.

Editor: Rubens Prates
Tradução: Rafael Zanolli
Revisão gramatical: Jeferson Ferreira
Revisão técnica: Fredy Gadotti
Editoração eletrônica: Camila Kuwabata e Carolina Kuwabata

ISBN: 978-85-7522-236-2

Histórico de impressões:

Abril/2011  Primeira edição

Novatec Editora Ltda.
Rua Luís Antônio dos Santos 110
02460-000 – São Paulo, SP – Brasil
Tel.: +55 11 2959-6529
Fax: +55 11 2950-8869
E-mail: novatec@novatec.com.br
Site: www.novatec.com.br
Twitter: twitter.com/novateceditora
Facebook: facebook.com/novatec
LinkedIn: linkedin.com/in/novatec

OGF20110323



20

capítulo 1

Criação de aplicações com  
desenvolvimento dirigido por testes

por Elad Elrom

No Flex 4, um dos objetivos é adotar um desenvolvimento de aplicação que seja 
mais centrado em design, e que separe a camada de apresentação da camada lógica, 
delegando mais responsabilidades ao designer que utiliza o Flash Catalyst. Diminuir 
as responsabilidades dos desenvolvedores permite que você se concentre na criação 
de aplicações Flash que sejam mais inteligentes, dinâmicas e complexas.

Todavia, conforme a aplicação Flash se torna mais complexa e dinâmica, e as necessi-
dades empresarias mudam rapidamente, mesmo dentro do próprio desenvolvimento, 
será sempre desafiador conseguir sustentar suas aplicações, ou torná-las escaláveis. O 
desafio é significativo e muitos desenvolvedores Flash descobrem que, mesmo utili-
zando um framework, o processo não se torna mais fácil. Tais desafios são comuns a 
qualquer tipo de desenvolvimento: para dispositivos móveis, web ou desktop.

Considere o seguinte: uma aplicação necessita de alterações, em razão de novas 
necessidades de negócio. Como você pode ter certeza de que as pequenas mudanças 
feitas não “quebram” outras partes da aplicação? Como pode ter certeza de que seu 
código está seguro, principalmente se não foi você quem o escreveu?

Tal problema não é novidade para engenheiros de software; desenvolvedores Java e 
ASP são diariamente desafiados pelas mesmas questões, e descobrem que o desen-
volvimento dirigido por testes (Test-Driven Development, ou TDD) é a melhor forma 
de criar aplicações de fácil manutenção.

O Flash se desenvolveu muito, desde sua origem, como uma pequena ferramenta de 
animação, até se tornar uma linguagem de programação “real”. Por isso, metodolo-
gias comuns em outras linguagens de programação também são necessárias para a 
criação de grandes aplicações dinâmicas. De fato, a Adobe e muitas outras empresas 
descobriram que a utilização da técnica TDD soluciona muitos dos desafios que os 
desenvolvedores encontram em seus ciclos de desenvolvimento diários.



21Capítulo 1 ■ Criação de aplicações com desenvolvimento dirigido por testes

Pessoalmente, percebo que muitos desenvolvedores conhecem essa técnica; ainda 
assim, relutam em utilizá-la por não saberem ao certo como fazê-lo, e por temerem 
que seu uso prolongue o tempo de desenvolvimento.

Em minha experiência pessoal, vejo que a utilização correta do TDD não aumenta o 
tempo gasto no desenvolvimento, mas acaba por reduzi-lo, além de facilitar a manuten-
ção da aplicação. Também descobri que é possível implementar o TDD em aplicações 
existentes, bem como aplicar seus métodos a muitos dos frameworks disponíveis, 
como o Cairngorm, ou o Robotlegs, de uma forma ou de outra.

O TDD é aplicável até mesmo em ambientes nos quais há um departamento de ga-
rantia de qualidade (Quality Assurance, ou QA), pois é preferível, sempre, entregar 
um código mais sólido, e que permita ao QA se concentrar na interface do usuário 
(User Interface, ou UI) e na criação dos casos de uso passíveis de testes.

Neste capítulo, você terá acesso a informações básicas com as quais poderá dar seus 
primeiros passos. Também será apresentado a alguns tópicos mais avançados, que 
explicarão de que modo você pode começar a utilizar o TDD em aplicações existen-
tes, e verá, ainda, como criar o teste de unidade1 para classes mais complicadas, que 
incluem chamadas de serviço e lógica complexa.

Visão geral do FlexUnit 4
Antes de tratarmos do TDD, gostaríamos de oferecer uma visão geral do FlexUnit 4, 
pois essa será a ferramenta que você utilizará para a criação de seus testes.

Vejamos o início de tudo. Ainda em 2003, a iteration::two, uma empresa de consultoria 
adquirida pela Adobe, e que depois se tornou um dos seus principais “braços”, lançou 
um produto conhecido como AS2Unit. Pouco depois, a Adobe lançou o Flex 1, em 
2004, e o AS2Unit ressurgiu como FlexUnit.

Até o lançamento do Flash Builder 4, era preciso efetuar o download do arquivo 
SWC, e criar manualmente o caso de teste e a suíte de testes. Com seu lançamento, a 
Adobe adicionou um plug-in FlexUnit como parte dos muitos assistentes (wizards) 
e plug-ins que facilitam o trabalho com o Flash Builder. O plug-in, automaticamente, 
efetua muitas das tarefas que antes tinham de ser feitas manualmente, simplificando 
o processo de teste de unidade. O projeto foi previamente lançado como FlexUnit 
0.9. Nunca recebeu, nem receberá, um lançamento 1.0 oficial, mas é conhecido como 
FlexUnit 1.

A nova versão, FlexUnit 4, é a mais próxima, em funcionalidade, do projeto JUnit 
(http://www.junit.org/), e aceita muitos dos recursos que ele oferece, além de outros 
1  N.T.: teste de unidade é um método pelo qual unidades individuais de código-fonte são testadas, 

para determinar se estão prontas para utilização. Uma unidade é a menor parte testável de uma 
aplicação (fonte: Wikipédia).

mais. O FlexUnit 4 combina recursos dos projetos prévios FlexUnit 1.0 e Fluint (http://
code.google.com/p/fluint/).

Alguns dos recursos principais do FlexUnit 4 são:

•	 maior	facilidade	na	criação	da	suíte	de	testes	e	das	classes	de	caso	de	teste;

•	 maior	facilidade	na	criação	de	test	runners	e	na	integração	com	os	runners	de	
outros frameworks;

•	 melhor	utilização	com	integração	contínua;

•	 melhor	manipulação	de	testes	assíncronos;

•	 melhor	manipulação	de	exceções;

•	 framework	orientado	por	metadados;

•	 permite	a	realização	de	testes	da	interface	do	usuário;

•	 habilidade	de	criar	sequências	de	testes.

Criando sua primeira suite de testes
1. Abra o Flash Builder 4. Selecione File ➤ New ➤ Flex Project. Dê ao projeto o nome 

de FlexUnit4 e selecione Finish.

O exemplo completo, que mostraremos nesta seção, pode ser baixado no site  
http://www.friendsofed.com/download.html?isbn=1430224835.

2. Clique no nome do projeto, para que o pacote seja selecionado automaticamente 
como o flexUnitTests default, e selecione File ➤ New ➤ Test Suite Class (Figura 1.1).

Figura 1.1 – Menu de criação de nova classe de suite de teste (Test Suite Class).

Na próxima janela, você pode definir o nome e incluir os testes. Dê à suíte o nome de 
FlexUnit4Suite, e pressione Finish. Perceba que você pode selecionar a versão do FlexUnit. 
Utilize a FlexUnit 4 (Figura 1.2).

Figura 1.2 – Janela New Test Suite Class .

Uma suíte de testes é uma composição de testes, que roda um grupo de casos de testes. 
Durante o desenvolvimento, você cria uma coleção de testes, que são empacotados em 
uma suíte, que pode ser rodada para garantir que seu código continue funcionando 
corretamente, mesmo depois de feitas algumas alterações.

O assistente criou uma pasta flexUnitTests, e uma classe FlexUnit4Suite.as (Figura 1.3).

Figura 1.3 – Package explorer mostrando a suite de testes que foi criada.

Abra a classe FlexUnit4Suite, que você criou.

package flexUnitTests {
 [Suite]
 [RunWith("org.flexunit.runners.Suite")]
 public class FlexUnit4Suite {
 }
}

Note que você está utilizando os metadados da suíte, que indicam que a classe 
é uma suíte. A tag Run With está utilizando o FlexUnit 4, para indicar qual runner 
será utilizado para executar o código.

O FlexUnit 4 é um grupo de runners que serão executados para criar uma série 
completa de testes. Você pode definir cada runner para implementar uma interface 
específica. Por exemplo, pode escolher utilizar a classe a que fez referência para rodar 
os testes, ao invés do runner default, integrado ao FlexUnit 4.

[RunWith("org.flexunit.runners.Suite")]
[RunWith("org.flexunit.experimental.theories.Theories")]

Isso significa que o framework é suficientemente flexível para aceitar runners futu-
ros, o que permite que os desenvolvedores criem seus próprios runners, utilizando a 
mesma UI. De fato, atualmente há runners que existem para o FlexUnit 1, FlexUnit 4, 
Fluint e SLT.

Criando sua primeira classe de caso de teste
O próximo passo é a criação da classe de caso de teste. Um caso de teste é composto 
pelas condições que você deseja expressar para verificar um requisito de negócio. 
Cada caso de teste, no FlexUnit 4, deve ser conectado a uma classe.

1. Selecione File ➤ New ➤ Test Case Class. Para o nome, escolha FlexUnitTester ➤ flexUnitTests. 
Clique em Finish (Figura 1.4). 

O assistente criou FlexUnitTester.as automaticamente, como mostra o código seguinte:

package flexUnitTests {
 public class FlexUnitTester {
  [Before]
  public function setUp():void {
  }

  [After]
  public function tearDown():void {
  }

  [BeforeClass]
  public static function setUpBeforeClass():void {
  }

  [AfterClass]
  public static function tearDownAfterClass():void {
  }
 }
}

Figura 1.4 – Janela New Test Case Class.

Perceba que, na janela do caso de teste, havia uma opção para anexar uma classe a ser 
testada (Figura 1.4). É bom ter essa opção disponível, caso você esteja integrando testes 
de unidade a código existente. Caso não esteja utilizando uma classe existente, mas 
seja possível que queira testar uma delas com o FlexUnit 4 (o que não corresponde 
exatamente ao TDD), você pode fazê-lo selecionando Next na janela New TestCase Class (ao 
invés de Finish). Na página seguinte, selecione os métodos a serem testados (Figura 1.5).

Você deve anexar o caso de teste à Test Suite criada previamente. Para fazê-lo, adicione 
uma referência, como a do código seguinte:

package flexUnitTests {
 [Suite]
 [RunWith("org.flexunit.runners.Suite")]
 public class FlexUnit4Suite {
  public var flexUnitTester:FlexUnitTester;
 }
}

Agora, pode rodar o teste. Selecione o ícone Run, e depois escolha FlexUnit Tests, no menu 
(Figura 1.6).



Flex 4 Avançado26

Figura 1.5 – Janela New Test Case Class.

Figura 1.6 – Rode a opção FlexUnit Tests.

Na janela seguinte, você pode selecionar as suítes de testes e os casos de testes. Em 
seu caso, há somente uma suíte para escolher, e nenhum caso, pois você ainda não 
criou nenhum teste em sua classe de Test Case (Figura 1.7).

Revise os resultados
O Flash Builder abre uma nova janela na qual os testes são executados, e os resultados 
são exibidos (Figura 1.8).

•	 1	teste	total	foi	executado;

•	 0	teve	sucesso;

•	 0	fracassou;

•	 1	apresentou	erro;

•	 0	foi	ignorado.

Figura 1.7 – Janela do assistente Run FlexUnit Tests.

Figura 1.8 – Resultados do FlexUnit no navegador.

Há um teste que apresentou erro. Feche o navegador e examine a janela de resultados 
do FlexUnit (Figura 1.9). O motivo de o teste ter falhado foi não haver métodos a serem 
testados, uma vez que você ainda não os criou.

Figura 1.9 – Janela FlexUnit Results.

A visualização dos resultados oferece diversos botões de ação que merecem destaque 
(Figura 1.10).

Figura 1.10 – Botões de ação da visualização dos resultados.

Substitua o código existente em FlexUnitTester.as pelo código seguinte. Explicaremos 
esse código detalhadamente na próxima seção:

package flexUnitTests {
 import flash.display.Sprite;
 import flexunit.framework.Assert;

 public class FlexUnitTester {
  //--------------------------------------------------------------------------
  //
  // Antes (Before) e Depois (After)
  //
  //--------------------------------------------------------------------------

  [Before]
  public function runBeforeEveryTest():void {
   // implemente
  }

  [After]
  public function runAfterEveryTest():void {
   // implemente
  }

  //--------------------------------------------------------------------------
  //
  // Testes
  //
  //--------------------------------------------------------------------------

  [Test]
  public function checkMethod():void {
   Assert.assertTrue( true );
  }

  [Test(expected="RangeError")]
  public function rangeCheck():void {
   var child:Sprite = new Sprite();
   child.getChildAt(0);
  }

  [Test(expected="flexunit.framework.AssertionFailedError")]
  public function testAssertNullNotEqualsNull():void {
   Assert.assertEquals( null, "" );
  }

  [Ignore("Not Ready to Run")]
  [Test]
  public function methodNotReadyToTest():void {
   Assert.assertFalse( true );
  }
 }
}

Rode o teste novamente e veja os resultados (Figura 1.11). Parabéns! Você criou sua 
primeira Test Suite, seu primeiro Test Case, e recebeu uma luz verde, indicando que 
o teste passou corretamente.

Figura 1.11 – Janela FlexUnit Results.

O FlexUnit 4 tem como base tags de metadados. Veja algumas das tags comuns a seguir:

•	 [Suite]: indica que a classe é uma Suite;

•	 [Test]: metadados Test substituem o texto prefixado à frente de cada método. 
Aceita os atributos expected, async, order, timeout, e ui;

•	 [RunWith]: seleciona o runner a ser utilizado;

•	 [Ignore]: em vez de eliminar um método com comentários, basta posicionar o 
metadado [Ignore];

•	 [Before]: substitui o método setup() no FlexUnit 1, e permite a utilização de 
métodos múltiplos. Aceita os atributos async, timeout, order, e ui;

•	 [After]: substitui o método teardown(), no FlexUnit 1, e permite a utilização de 
métodos múltiplos. Aceita os atributos async, timeout, order, e ui;

•	 [BeforeClass]: permite a execução de métodos antes da classe de teste. Aceita o 
atributo order;

•	 [AfterClass]: permite a execução de métodos depois da classe de teste. Aceita o 
atributo order.

Como podemos ver no exemplo, você utilizou muitas tags, como RangeError, 
AssertionFailedError, e Ignore. O uso dos metadados facilita a criação do código. Expli-
caremos o código em si na seção seguinte.

Métodos de asserção

Voltando ao exemplo: as tags de metadados Before e After indicam que esses métodos 
serão executados antes e depois de cada método de teste.

[Before]
public function runBeforeEveryTest():void {
 // implemente
}

[After]
public function runAfterEveryTest():void {
 // implemente
}

O metadado Test substitui o prefixo à frente de cada método, para que você tenha 
um método que não inicie com test. Perceba que, diferentemente da versão prévia 
do FlexUnit, você tem de indicar a classe a ser afirmada, uma vez que sua classe não 
herda mais de TestCase.

[Test]
public function checkMethod():void {
 Assert.assertTrue( true );
}

No FlexUnit 1, era comum eliminar com comentários os métodos que você não queria 
mais testar. A tag de metadados Ignore, no FlexUnit 4, permite que se ignore um método.

[Ignore("Not Ready to Run")]
[Test]
public function methodNotReadyToTest():void {
 Assert.assertFalse( true );
}

Caso queira criar uma ordem para os testes, ou para os métodos before e after, 
basta adicionar o atributo order.

 [Test(order=1)]

Há muitos outros métodos de asserção que podem ser utilizados na criação de testes 
(Tabela 1.1).

Tabela 1.1 – Descrição e métodos de classe das asserções

Tipo da asserção Descrição

assertEquals Afirma que dois valores são iguais.

assertContained Afirma que a primeira string está contida na segunda.

assertNoContained Afirma que a primeira string não está contida na segunda.

assertFalse Afirma que uma condição é falsa.

assertTrue Afirma que uma condição é verdadeira.

assertMatch Afirma que uma string corresponde a uma regexp.

assertNoMatch Afirma que uma string não corresponde a uma regexp.

assertNull Afirma que um objeto é nulo.

assertNotNull Afirma que um objeto não é nulo.

assertDefined Afirma que um objeto é definido.

assertUndefined Afirma que um objeto é indefinido.

assertStrictlyEquals Afirma que dois objetos são estritamente idênticos.

assertObjectEquals Afirma que dois objetos são iguais.

Para utilizar um método de asserção, passe uma string de mensagem e dois parâme-
tros a serem comparados. A string é a mensagem que será utilizada, caso o teste falhe.

assertEquals("Error testing the application state", state, 1);

Se você omitir a string de mensagem, receberá a mensagem default. No editor, digite 
Assert, para ter acesso às dicas de código para os métodos de asserção disponíveis 
no IDE do FlashBuilder.

Manipulação de exceções

Os metadados de teste permitem a definição de um atributo de exceção, e tornam 
possível o teste de exceções. Isso funciona da seguinte forma: o método de teste, com 
os atributos esperados, aponta para a mensagem de erro que você espera, e o teste 
passará quando a exceção for levantada.

O exemplo seguinte demonstra o atributo esperado dos metadados Test. O método 
rangeCheck cria um novo objeto Sprite. O código produzirá um teste de sucesso, pois o 
filho no índice 1 não existe, fazendo com que o código provoque uma exceção durante 
o tempo de execução.

[Test(expected="RangeError")]
public function rangeCheck():void {
 var child:Sprite = new Sprite();
 child.getChildAt(0);
}

Outro exemplo pode ser observado quando esperamos um erro de asserção. Veja o 
método testAssertNullNotEqualsNull. O método está esperando o erro AssertionFailedError. 
O método assertEquals terá sucesso, já que nulo é igual a nulo, fazendo com que o teste 
falhe. Altere a instrução para a seguinte:

Assert.assertEquals( null, null );

Dessa maneira, fará com que o teste tenha sucesso.

[Test(expected="flexunit.framework.AssertionFailedError")]
public function testAssertNullNotEqualsNull():void {
 Assert.assertEquals( null, null );
}

Test runners

O test runner é semelhante ao test runner do FlexUnit 1. Veja o código completo no 
exemplo seguinte.

<?xml version="1.0" encoding="utf-8"?>
<s:Application xmlns:fx="http://ns.adobe.com/mxml/2009"
 xmlns:s="library://ns.adobe.com/flex/spark"
 xmlns:mx="library://ns.adobe.com/flex/halo"
 xmlns:flexUnitUIRunner="http://www.adobe.com/2009/flexUnitUIRunner"
 minWidth="1024" minHeight="768" creationComplete="creationCompleteHandler(event)">

 <fx:Script>
  <![CDATA[
   import flexUnitTests.FlexUnit4TheorySuite;
   import flexUnitTests.FlexUnit4HamcrestSuite;
   import mx.events.FlexEvent;
   import flexUnitTests.FlexUnit4Suite;

   import org.flexunit.listeners.UIListener;
   import org.flexunit.runner.FlexUnitCore;

   // armazena uma instância da classe <code>FlexUnitCore</code>
   private var flexUnitCore:FlexUnitCore;

   protected function creationCompleteHandler(event:FlexEvent):void {
    flexUnitCore = new FlexUnitCore();

    // ouvinte (listener) para a UI, opcional
    flexUnitCore.addListener( new UIListener( testRunner ));

    // esta instrução run executa os testes de unidade para o framework FlexUnit4
    flexUnitCore.run( FlexUnit4Suite, FlexUnit4HamcrestSuite,FlexUnit4TheorySuite );
   }
  ]]>
 </fx:Script>

 <flexUnitUIRunner:TestRunnerBase id="testRunner" width="100%" height="100%" />

</s:Application>

Você criou uma instância de FlexUnitCore, que estende EventDispatcher e atua como 
fachada para a execução dos testes, além de ser semelhante à JUnitCore em métodos 
e funcionalidade.

// armazena uma instância da classe <code>FlexUnitCore</code>
private var flexUnitCore:FlexUnitCore;

Assim que o manipulador creationCompleteHandler for chamado, você pode adicionar o 
ouvinte UIListener:

protected function creationCompleteHandler(event:FlexEvent):void {
 flexUnitCore = new FlexUnitCore();

Você tem a opção de definir o ouvinte para a UI. O que ocorre é que a classe FlexUnitCore 
tem um observador assíncrono, que aguarda até que todos os ouvintes estejam con-
cluídos, antes de iniciar a execução do runner.

// ouvinte (listener) para a UI, opcional
flexUnitCore.addListener( new UIListener( testRunner ));

A linha seguinte chama o método run, para iniciar a execução dos testes. O método 
run aceita ...args como parâmetros, e assim você será capaz de listar todas as suítes 
que deseja que o runner execute.

 // essa instrução run executa os testes de unidade para o framework FlexUnit4
 flexUnitCore.run( FlexUnit4Suite );
}

Você também precisa de uma instância do componente TestRunnerBase.

<flexUnitUIRunner:TestRunnerBase id="testRunner" width="100%" height="100%" />

Compile e rode os testes. Depois, poderá ver os resultados.

Método de asserção Hamcrest

Além das novas asserções-padrão, como Assert.assertEquals ou Assert.assertFalse, o 
FlexUnit 4 aceita novos métodos, graças à Hamcrest (http://github.com/drewbourne/
hamcrest-as3/tree/master), uma biblioteca que tem como base a noção de matchers 
(correspondentes). Cada matcher pode ser definido para corresponder a condições 
para suas asserções.

Crie um novo caso de teste, FlexUnitCheckRangeTester.

package flexUnitTests {
 import org.hamcrest.AbstractMatcherTestCase;
 import org.hamcrest.number.between;

 public class FlexUnitCheckRangeTester extends AbstractMatcherTestCase {
  //--------------------------------------------------------------------------
  //
  // Antes (Before) e depois (After)
  //
  //--------------------------------------------------------------------------

  private var numbers:Array;

  [Before]
  public function runBeforeEveryTest():void {
   numbers = new Array( 0, 1, 2, 3, 4, 5, 6, 7, 8, 9, 10 );
  }

  [After]
  public function runAfterEveryTest():void {
   numbers = null;
  }

  //--------------------------------------------------------------------------
  //
  // Testes
  //
  //--------------------------------------------------------------------------

  [Test]
  public function betweenRangeExclusive():void {
   assertMatches("assert inside range", between(numbers[0], numbers[10], true), 5);
   assertDoesNotMatch("assert outside range", between (numbers[0], numbers[10], true), 11);
  }

  [Test]
  public function betweenRangeInclusive():void {
   assertMatches("assert inside range", between(numbers[0], numbers[10]), 5);

   assertDoesNotMatch("assert outside range", between(numbers[0], numbers[10]), 11);
  }

  [Test]
  public function betweenReadableDescription():void {
   assertDescription("a Number between <0> and <10>", between (numbers[0], numbers[10]));
   assertDescription("a Number between <0> and <10> exclusive", betweenÉ
    (numbers[0], numbers[10], true));
  }
 }
}

A classe FlexUnit4HamcrestSuite inclui a classe FlexUnitCheckRangeTester.

package flexUnitTests {

 [Suite]
 [RunWith("org.flexunit.runners.Suite")]
 public class FlexUnit4HamcrestSuite {
  public var flexUnitCheckRangeTester:FlexUnitCheckRangeTester;
 }
}

Por fim, não se esqueça de incluir a nova suíte de testes em FlexUnit4.mxml:

flexUnitCore.run( FlexUnit4Suite, FlexUnit4HamcrestSuite );

Testes assíncronos

Caso você tenha trabalhado com o FlexUnit 1 no passado, sabe que nem sempre foi 
uma tarefa fácil criar testes assíncronos, e testar código orientado a eventos. Muitas 
vezes, precisávamos modificar classes existentes, apenas para acomodar o FlexUnit, 
ou, ainda, tínhamos de criar testes utilizando diversos hacks2. Uma das maiores 
vantagens do Fluint é a habilidade de acomodar múltiplos eventos assíncronos. O 
FlexUnit 4 incorporou a funcionalidade do Fluint, e aceita testes assíncronos avan-
çados, incluindo setups e teardowns assíncronos.

Como exemplo, crie um novo Test Case, TestAsynchronous.

package flexUnitTests {
 import flash.events.Event;
 import flash.events.EventDispatcher;

 import flexunit.framework.Assert;

 import mx.rpc.events.FaultEvent;
 import mx.rpc.events.ResultEvent;

2  N.T.: refere-se a uma solução engenhosa, ou rápida, para um problema de programação, ou, 
ainda, a uma solução que pode ser vista como desajeitada, ou deselegante – mas que, em geral, 
é relativamente rápida (fonte: Wikipédia).

 import mx.rpc.http.HTTPService;

 import org.flexunit.async.Async;

 public class AsynchronousTester {
  private var service:HTTPService;
  //--------------------------------------------------------------------------
  //
  // Antes (Before) e depois (After)
  //
  //--------------------------------------------------------------------------

  [Before]
  public function runBeforeEveryTest():void {
   service = new HTTPService();
   service.resultFormat = "e4x";
  }

  [After]
  public function runAfterEveryTest():void {
   service = null;
  }
  //--------------------------------------------------------------------------
  //
  // Testes
  //
  //--------------------------------------------------------------------------

  [Test(async,timeout="500")]
  public function testServiceRequest():void {
   service.url = "../assets/file.xml";
   service.addEventListener(ResultEvent.RESULT, Async.asyncHandlerÉ
    ( this, onResult, 500 ), false, 0, true );
   service.send();
  }

  [Test(async,timeout="500")]
  public function testeFailedServicRequest():void {
   service.url = "file-that-dont-exists";
   service.addEventListener( FaultEvent.FAULT, Async.asyncHandlerÉ
    ( this, onFault, 500 ), false, 0, true );
   service.send();
  }

  [Test(async,timeout="500")]
  public function testEvent():void {
   var EVENT_TYPE:String = "eventType";

   var eventDispatcher:EventDispatcher = new EventDispatcher();
   eventDispatcher.addEventListener(EVENT_TYPE, Async.asyncHandlerÉ
    ( this,  handleAsyncEvnet, 500 ), false, 0, true );
   eventDispatcher.dispatchEvent( new Event(EVENT_TYPE));
  }

  [Test(async,timeout="500")]
  public function testMultiAsync():void  {
   testEvent();
   testServiceRequest();
  }

  //--------------------------------------------------------------------------
  //
  // Manipuladores assíncronos
  //
  //--------------------------------------------------------------------------

  private function onResult(event:ResultEvent, passThroughData:Object):void {
   Assert.assertTrue( event.hasOwnProperty("result"));
  }

  private function handleAsyncEvnet(event:Event, passThroughData:Object):void {
   Assert.assertEquals( event.type, "eventType" );
  }

  private function onFault(event:FaultEvent, passThroughData:Object):void {
   Assert.assertTrue( event.fault.hasOwnProperty("faultCode"));
  }
 }
}

Os metadados Before e After serão executados antes de cada método de teste. Como 
você está realizando um teste assíncrono, pode definir service como null, quando 
tiver terminado, para evitar memory leaks3.

[Before]
public function runBeforeEveryTest():void {
 service = new HTTPService();
 service.resultFormat = "e4x";
}
[After]

3  N.T.: memory leak é o nome dado a um tipo específico de consumo de memória por um pro-
grama de computador, no qual o programa é incapaz de liberar a memória que adquire. Um 
memory leak apresenta sintomas semelhantes a diversos outros problemas e, geralmente, só pode 
ser diagnosticado por um programador com acesso ao código-fonte do programa. Entretanto, 
muitas pessoas definem memory leak como qualquer tipo de acréscimo indesejado na utilização 
da memória, mesmo que tal apontamento não seja estritamente correto (fonte: Wikipédia).



Flex 4 Avançado38

public function runAfterEveryTest():void {
 service = null;
}

Você pode definir o teste com os metadados assíncronos para realizar um teste as-
síncrono, e definir o timeout em 500 milissegundos. Assim que enviar a requisição 
para recuperar um arquivo XML, a requisição de resultado irá para o método onResult.

 [Test(async,timeout="500")]
public function testServiceRequest():void {
 service.url = "../assets/file.xml";
 service.addEventListener(ResultEvent.RESULT, Async.asyncHandlerÉ
  ( this, onResult, 500 ), false, 0, true );
 service.send();
}

O manipulador de resultado para o stub4 de teste testServiceRequest utiliza Assert.
assertTrue para garantir que há resultados na requisição que você definiu. Event tem 
uma propriedade de "result", por isso a expressão resultará em um valor Booleano 
verdadeiro.

private function onResult(event:ResultEvent, passThroughData:Object):void {
 Assert.assertTrue( event.hasOwnProperty("result"));
}

O segundo teste verificará requisições falhas. Estamos, propositadamente, definindo 
um URL que não existe, "file-that-dont-exists", o que resultará em uma chamada de 
serviço falha. O método de resultado onFault espera que o objeto de evento tenha o 
parâmetro faultCode.

 [Test(async,timeout="500")]
 public function testeFailedServicRequest():void {
  service.url = "file-that-dont-exists";
  service.addEventListener( FaultEvent.FAULT, Async.asyncHandlerÉ
   ( this, onFault, 500 ), false, 0, true );
  service.send();
 }

private function onFault(event:FaultEvent, passThroughData:Object):void {
 Assert.assertTrue( event.fault.hasOwnProperty("faultCode"));
}

No stub de teste testEvent, você está testando e despachando um evento, e esperando o 
resultado depois de 500 milissegundos. Definimos o tipo do evento para “eventType”.

4  N.T.: um stub, ou method stub, em português "esboço de método", em desenvolvimento de 
software, é um fragmento de código usado para substituir algumas outras funcionalidades 
de programação. Um stub pode simular o comportamento de um código existente, ou ser um 
substituto temporário para código ainda a ser desenvolvido (fonte: Wikipédia).

 [Test(async,timeout="500")]
 public function testEvent():void {
  var EVENT_TYPE:String = "eventType";
  var eventDispatcher:EventDispatcher = new EventDispatcher();

  eventDispatcher.addEventListener(EVENT_TYPE, Async.asyncHandlerÉ
   ( this,  handleAsyncEvnet, 500 ), false, 0, true );
  eventDispatcher.dispatchEvent( new Event(EVENT_TYPE));
 }

 [Test(async,timeout="500")]
 public function testMultiAsync():void  {
  testEvent();
  testServiceRequest();
 }

No manipulador de resultado, você assegura que o tipo do evento é igual ao definido 
na requisição, o que significa que é o correto.

 private function handleAsyncEvnet(event:Event, passThroughData:Object):void {
  Assert.assertEquals( event.type, "eventType" );
 }

Não se esqueça de ajustar FlexUnit4Suite.as para incluir a classe de Test Case criada 
antes de rodar o teste.

package flexUnitTests {
 [Suite]
 [RunWith("org.flexunit.runners.Suite")]
 public class FlexUnit4Suite {
  public var flexUnitTester:FlexUnitTester;
  public var asynchronousTester:AsynchronousTester;
 }
}

ActionScript 3 é uma linguagem orientada a eventos e, muitas vezes, é preciso testar 
casos assíncronos. Como mostraram os exemplos, é fácil criar testes assíncronos no 
FlexUnit 4, e garantir que eles sejam realizados.

Teorias

O FlexUnit 4 introduziu um conceito totalmente novo: teorias. Uma teoria, como 
sugere o nome, permite que você crie um teste para verificar suas suposições no que se 
refere a como um teste deve se comportar. Esse tipo de teste é útil quando você precisa 
verificar um método com um número grande, ou até mesmo infinito, de conjuntos 
de valores. O teste estabelece parâmetros (pontos de dados, ou data points), e esses 
pontos de dados podem ser utilizados em conjunção com o teste.

Crie uma nova suíte de testes, e dê a ela o nome de FlexUnit4TheorySuite.

package flexUnitTests {
 import org.flexunit.assertThat;
 import org.flexunit.assumeThat;
 import org.flexunit.experimental.theories.Theories;
 import org.hamcrest.number.greaterThan;
 import org.hamcrest.object.instanceOf;

 [Suite]
 [RunWith("org.flexunit.experimental.theories.Theories")]
 public class FlexUnit4TheorySuite {
  private var theory:Theories;

  //--------------------------------------------------------------------------
  //
  // DataPoints
  //
  //--------------------------------------------------------------------------

  [DataPoint]
  public static var number:Number = 5;

  //--------------------------------------------------------------------------
  //
  // Teorias
  //
  //--------------------------------------------------------------------------

  [Theory]
  public function testNumber( number:Number ):void {
   assumeThat( number, greaterThan( 0 ));
   assertThat( number, instanceOf(Number));
  }
 }
}

A tag RunWith aponta para um runner que implementa uma interface diferente da 
default.

 [Suite]
 [RunWith("org.flexunit.experimental.theories.Theories")]

Definimos um ponto de dado com um número.

 [DataPoint]
 public static var number:Number = 5;

Em seguida, definimos a teoria que gostaríamos de verificar. Nosso teste confere, 
verificando que o número é maior que cinco, e que é do tipo Number.

 [Theory]
 public function testNumber( number:Number ):void {
  assumeThat( number, greaterThan( 0 ));
  assertThat( number, instanceOf(Number));
 }

Adicione a suíte de teste ao FlexUnit4.mxml:

flexUnitCore.run( FlexUnit4Suite, FlexUnit4HamcrestSuite, FlexUnit4TheorySuite );

Testando interfaces de usuário

O Flex é utilizado para criar GUIs, e inclui aparência e comportamento visuais. Às 
vezes, é possível que você queira testar o visual e o comportamento de sua aplicação.

No FlexUnit 1, não havia nenhuma habilidade de teste de interfaces de usuário, 
e componentes MXML não eram uma opção na criação dos testes de unidade. O 
FlexUnit 4 inclui o conceito de sequência, permitindo que você crie sequências que 
incluam todas as operações que deseja efetuar na UI.

Por exemplo, vamos supor que você queira testar o clique de um usuário em um 
botão de sua aplicação. Veja o código seguinte:

package flexUnitTests {
 import flash.events.Event;
 import flash.events.MouseEvent;

 import mx.controls.Button;
 import mx.core.UIComponent;
 import mx.events.FlexEvent;

 import org.flexunit.Assert;
 import org.flexunit.async.Async;
 import org.fluint.sequence.SequenceCaller;
 import org.fluint.sequence.SequenceEventDispatcher;
 import org.fluint.sequence.SequenceRunner;
 import org.fluint.sequence.SequenceSetter;
 import org.fluint.sequence.SequenceWaiter;
 import org.fluint.uiImpersonation.UIImpersonator;

 public class FlexUnit4CheckUITester {
  private var component:UIComponent;
  private var btn:Button;

  //--------------------------------------------------------------------------
  //
  // Antes (Before) e depois (After)
  //
  //--------------------------------------------------------------------------

  [Before(async,ui)]
  public function setUp():void {
   component = new UIComponent();
   btn = new Button();
   component.addChild( btn );
   btn.addEventListener( MouseEvent.CLICK, function():voidÉ
    { component.dispatchEvent( new Event( 'myButtonClicked' )); } )

   Async.proceedOnEvent( this, component, FlexEvent.CREATION_COMPLETE, 500 );
   UIImpersonator.addChild( component );
  }

  [After(async,ui)]
  public function tearDown():void {
   UIImpersonator.removeChild( component );
   component = null;
  }

  //--------------------------------------------------------------------------
  //
  // Testes
  //
  //--------------------------------------------------------------------------

  [Test(async,ui)]
  public function testButtonClick():void {
   Async.handleEvent( this, component, "myButtonClicked", handleButtonClickEvent, 500);
   btn.dispatchEvent( new MouseEvent( MouseEvent.CLICK, true, false ));
  }

  [Test(async,ui)]
  public function testButtonClickSequence():void {
   var sequence:SequenceRunner = new SequenceRunner( this );

   var passThroughData:Object = new Object();
   passThroughData.buttonLable = 'Click button';

   with ( sequence ) {
    addStep( new SequenceSetter( btn, {label:passThroughData.buttonLable} ));
    addStep( new SequenceWaiter( component, 'myButtonClicked', 500 ));
    addAssertHandler( handleButtonClickSqEvent, passThroughData );

    run();
   }

   btn.dispatchEvent( new MouseEvent( MouseEvent.CLICK, true, false ));
  }

  //--------------------------------------------------------------------------
  //
  // Manipuladores
  //
  //--------------------------------------------------------------------------
  private function handleButtonClickEvent( event:Event, passThroughData:Object ):void {
   Assert.assertEquals( event.type, "myButtonClicked" );
  }
  private function handleButtonClickSqEvent( event:*, passThroughData:Object ):void {
   Assert.assertEquals(passThroughData.buttonLable, btn.label);
  }
 }
}

Você armazena as instâncias do componente e do botão que testará. Isso é apenas um 
exemplo, mas, em testes reais da UI, você utilizará o componente MXML. É possível 
criar uma instância de um componente ou aplicação MXML, da forma como vemos 
neste exemplo.

 private var component:UIComponent;
 private var btn:Button;

A tag Before tem os atributos async e UI, para indicar que você aguardará pelo evento 
FlexEvent.CREATION_COMPLETE. Uma vez recebido o evento, ele adicionará o componente 
que você criou ao componente UIImpersonator, que estende a classe Assert, e permite a 
adição de componentes e a realização de testes.

No método setUp(), você adiciona um botão e define um manipulador de eventos, 
quando ele é clicado. Em seu caso, você despachará o evento myButtonClicked.

 [Before(async,ui)]
 public function setUp():void {
  component = new UIComponent();
  btn = new Button();
  component.addChild( btn );
  btn.addEventListener( MouseEvent.CLICK, function():voidÉ
   { component.dispatchEvent( new Event( 'myButtonClicked' )); } )
  Async.proceedOnEvent( this, component, FlexEvent.CREATION_COMPLETE, 500 );
  UIImpersonator.addChild( component );
 }

Depois de concluir cada teste, você removerá component do UIImpersonator e definirá 
component como null.

 [After(async,ui)]
 public function tearDown():void {
  UIImpersonator.removeChild( component );
  component = null;
 }

O primeiro teste que você criará é simples, e serve para verificar o que ocorre quando 
o botão é pressionado. Lembre-se de que levantamos um evento myButtonClicked. Uma 
vez despachado o evento, o manipulador chamará o método handleButtonClickEvent.

 [Test(async,ui)]
 public function testButtonClick():void {
  Async.handleEvent( this, component, "myButtonClicked", handleButtonClickEvent, 500 );
  btn.dispatchEvent( new MouseEvent( MouseEvent.CLICK, true, false ));
 }

No segundo teste, você criará uma sequência. Uma sequência permite que você tente 
imitar a utilização que o usuário faria do componente. Em nosso caso, testaremos o 
rótulo (label) e a ação de pressionar o botão. Note que você está definindo um obje-
to passThroughData que armazenará o nome do label. Você passará o nome do label e, 
depois, fará a comparação, quando o teste assíncrono estiver concluído.

 [Test(async,ui)]
 public function testButtonClickSequence():void {
  var sequence:SequenceRunner = new SequenceRunner( this );

  var passThroughData:Object = new Object();
  passThroughData.buttonLable = 'Click button';

  with ( sequence ) {
   addStep( new SequenceSetter( btn, {label:passThroughData.buttonLable} ));
   addStep( new SequenceWaiter( component, 'myButtonClicked', 500 ));
   addAssertHandler( handleButtonClickSqEvent, passThroughData );

   run();
  }

  btn.dispatchEvent( new MouseEvent( MouseEvent.CLICK, true, false ));
 }

O método handleButtonClickEvent é o manipulador para o método testButtonClick. Você 
está, apenas, verificando que o evento despachado é do tipo myButtonClicked.

private function handleButtonClickEvent( event:Event, passThroughData:Object ):void {
 Assert.assertEquals( event.type, "myButtonClicked" );
}

O segundo manipulador será utilizado pelo método testButtonClickSequence. Portanto, 
verifique que o label do botão sofreu alteração.

 private function handleButtonClickSqEvent( event:*, passThroughData:Object ):void {
  Assert.assertEquals(passThroughData.buttonLable, btn.label);
 }
}

Como mostra o exemplo precedente, agora podemos testar a aparência, assim como 
o comportamento visual no FlexUnit 4, o que pode ser de grande ajuda quando 
estivermos criando GUIs. Ainda assim, mesmo que você possa testar uma GUI com 
o FlexUnit 4, recomendamos a ferramenta de código aberto FlexMonkey (http://code.
google.com/p/flexmonkey/), para a realização de testes mais complexos.

Desenvolvimento dirigido por testes no FlexUnit 4
O FlexUnit e a prática do TDD caminham lado a lado. TDD é uma técnica de de-
senvolvimento de software, e o FlexUnit 4 pode ser utilizado para implementá-la. O 
desenvolvimento dessa metodologia teve início com programadores adicionando um 
teste aos códigos, depois de escrevê-lo. Em 1999, a metodologia Extreme Programming 
(XP) discutiu o desenvolvimento de projetos dinâmicos com diferentes requisitos e 
um ciclo de desenvolvimento que incluíssem TDD para criar o teste antes do código 
em si. Perceba que o TDD não é o ciclo de desenvolvimento completo, mas apenas 
parte do paradigma de desenvolvimento XP. Ao preparar os testes antes de escrever 
o código, o TDD permite que você exiba seu trabalho gradualmente, em vez de fazer 
com que o cliente espere pelo resultado completo.

O avanço em pequenos incrementos permite que seus clientes realizem as alterações 
que julgam necessárias antes da criação do código final, além de garantir que nada 
dê errado, e que seu código faça somente o necessário, nada mais. É importante men-
cionar que o foco da técnica TDD é produzir código, e não criar uma plataforma de 
testes. A habilidade de realizar testes é apenas um benefício adicional.

O TDD repousa sobre o conceito de que tudo que você cria deve ser testado, e, se 
não for capaz de fazê-lo, você deve pensar duas vezes se o que quer criar é, realmente, 
necessário.

A XP utiliza o conceito de técnicas TDD com um processo curto de desenvolvimento, 
a cada duas semanas: é o Iteration Planning (Planejamento de Iterações), que tem 
como base histórias de usuários (veja o exemplo seguinte). O planejamento de itera-
ções contém histórias de usuários e consiste no código necessário para passar pelo 
teste. Uma vez concluído o teste, o código é refatorado para remover código extra e 
criar um código mais limpo. Ao final do planejamento da iteração, a equipe entrega 
uma aplicação em funcionamento.

Uma história de usuário (user story) é utilizada para representar um requisito 
de negócio. O cliente explica o requisito ao engenheiro do software, e todos 
esses requisitos são histórias de usuário que, quando reunidas, constituem a 
especificação de requisitos. Por exemplo, “quando o usuário efetua o login, 
uma janela de boas-vindas deve surgir”.

Vamos explorar a técnica TDD (Figura 1.12).

1. Adicione um teste. O primeiro passo é compreender as necessidades de negócio 
(que se transformam em histórias de usuários) e pensar em todos os cenários 
possíveis. Caso os requisitos não sejam suficientemente claros, você pode le-
vantar antecipadamente novas questões, em vez de aguardar que o software 
esteja concluído, o que demandaria um maior Level of Effort5, ou LOE, para 
realizar uma mudança.

2. Escreva um teste de unidade falho. Essa fase é necessária para garantir que o teste de 
unidade, em si, funcione corretamente e não passe, já que você não escreveu 
código algum.

3. Escreva o código. Durante essa fase, você escreve o código da forma mais simples e 
eficiente possível, para garantir que ele passe. Não é necessário incluir padrões 
de projeto, pensar sobre o restante da aplicação, ou tentar limpar o código. Sua 
meta é apenas passar o teste.

4. Passe o teste. Uma vez escrito todo o código, o teste passa, e você sabe que ele 
atende a todos os requisitos de negócio. Então, pode compartilhar seu trabalho 
com o cliente, ou com outros membros de sua equipe.

5. Refatore. Agora que o teste está concluído, e que você confirmou que ele atende 
aos requisitos de negócio, pode garantir que o código está pronto para produ-
ção, substituindo parâmetros temporários, adicionando padrões de projeto, 
removendo duplicatas e criando classes para fazer o necessário. Preferivelmente, 
quando essa fase estiver concluída, o código será entregue para revisão, processo-
chave para garantir que o trabalho está satisfatório, e que obedece ao padrão 
de codificação da empresa.

6. Repita. O teste de unidade está concluído. Você pode avançar para o próximo 
teste de unidade e publicar seu código, para compartilhá-lo com clientes e 
outros membros de sua equipe.

Para começar a trabalhar com o Flash Builder 4, efetue o seguinte processo:

Abra o Flash Builder 4. Selecione File ➤ New ➤ Flex Project. Dê ao projeto o nome 
FlexUnitExample, e pressione ok.

6  N.T.: no gerenciamento de projetos, o Level of Effort (LOE) é uma atividade de suporte que deve 
ser feita para oferecer sustentação a outras atividades, ou ao esforço inteiro do projeto. Normal-
mente, consiste em pequenas porções de trabalho, que devem ser repetidas periodicamente (fonte: 
Wikipédia).

Figura 1.12 – Diagrama de processo para o desenvolvimento dirigido por teste.

Agora, você está pronto para começar. Tradicionalmente, muitos processadores exigem 
que você utilize um quadro, e crie um diagrama UML antes de iniciar o trabalho. 
Entretanto, criar o diagrama primeiro pode prolongar o tempo de desenvolvimento. 
Considere o seguinte exemplo: você recebe um projeto com requisitos de negócio, 
mas, como ainda não fez o trabalho, pode apenas supor como tudo será feito. Dessa 
forma, seu diagrama é dirigido de acordo com o que você acha que é necessário, ao 
invés do que realmente é necessário. O TDD vira esse processo de cabeça para baixo. 
Comece com seus requisitos de negócio:

•	 você	precisa	de	uma	classe	utilitária	que	seja	capaz	de	ler	um	arquivo	XML 
com informações do empregado;

•	 uma	vez	recuperada	as	informações	do	empregado,	elas	podem	ser	processadas	
em uma classe de objeto de valor;

•	 a	informação	será,	então,	exibida	em	uma	tela;

•	 tendo	em	mente	esses	requisitos,	você	pode	seguir	em	frente.

Criação da suíte de testes e do caso de teste

Seu próximo passo será a criação da Test Suite e do Test Case.

1. Clique no nome do projeto para que o pacote seja selecionado como o 
flexUnitTests default, e selecione File ➤ New ➤ Test Suite Class.



Flex 4 Avançado48

2. Na janela seguinte, você pode definir o nome e incluir os testes. Dê à suite o 
nome GetEmployeesSuite e pressione Finish.

3. Depois, crie a classe de caso de teste, escolhendo File ➤ New ➤ Test Case Class. Para 
o nome, escolha GetEmployeesInfoTester, e para a Class to test. Clique em Finish.

O código gerado procura por classes de Test Suites que você estará testando. Ainda 
que a suíte de testes tenha sido criada automaticamente, o flexUnitTests não inclui a 
classe Test Suite.

package flexUnitTests {
 [Suite]
 [RunWith("org.flexunit.runners.Suite")]
 public class GetEmployeesSuite {
 }
}

Lembre-se de definir uma referência a todos os testes que você combinará em sua 
suíte; por isso, adicione uma referência à classe. Veja o código seguinte como exemplo:

package flexUnitTests {
 [Suite]
 [RunWith("org.flexunit.runners.Suite")]
 public class GetEmployeesSuite {
  public var getEmployeesInfoTester:GetEmployeesInfoTester;
 }
}

FlexUnit 4, nos bastidores

Na estrutura de pastas da aplicação, você pode encontrar GetEmployeesInfoTest.as, 
GetEmployeesInfoSuite.as e flexUnitCompilerApplication.mxml (Figura 1.13).

Figura 1.13 – Package explorer do Flash Builder 4.

Na estrutura de pastas da aplicação, você pode encontrar os arquivos GetEmployee-
sInfoTester.as, GetEmployeesSuite.as e FlexUnitCompilerApplication.mxml (Figura 1.14).

Figura 1.14 – A estrutura de pastas de CalculatorApplication.

O pacote bin-debug/generated armazena todos os arquivos criados pelo com-
pilador MXMLC, e normalmente invisíveis para você. Para vê-los, selecione o 
projeto, clique com o botão direito, e selecione Properties. Sob Flex Compiler, em 
Additional Compiler Arguments, adicione a seguinte opção: -keep-generated-
actionscript, ou -keep-generated-actionscript=true.

Examine o FlexUnitApplication.mxml.

<?xml version="1.0" encoding="utf-8"?>
<!-- This is an auto generated file and is not intended for modification. -->
<s:Application xmlns:fx="http://ns.adobe.com/mxml/2009"
 xmlns:s="library://ns.adobe.com/flex/spark"
 xmlns:mx="library://ns.adobe.com/flex/mx" minWidth="955"É
  minHeight="600" xmlns:flexui="flexunit.flexui.*" creationComplete="onCreationComplete()">
 <fx:Script>
  <![CDATA[
   import flexUnitTests.GetEmployeesSuite;

   public function currentRunTestSuite():Array {
    var testsToRun:Array = new Array();
    testsToRun.push(flexUnitTests.GetEmployeesSuite);
    return testsToRun;
   }
   private function onCreationComplete():void {
    testRunner.runWithFlexUnit4Runner(currentRunTestSuite(), "FlexUnitExample");
   }
  ]]>



Flex 4 Avançado50

 </fx:Script>
 <fx:Declarations>
  <!-- Place non-visual elements (e.g., services, value objects) here -->
 </fx:Declarations>
 <flexui:FlexUnitTestRunnerUI id="testRunner">
 </flexui:FlexUnitTestRunnerUI>
</s:Application>

Como o leitor pode ver, há uma GUI, FlexUnitTestRunnerUI, semelhante em funcionali-
dade à classe FlexUnit TestRunner no FlexUnit 0.9. Essencialmente, a aplicação adiciona 
o teste inteiro e o roda na UI.

No FlexUnitApplication.mxml, você pode ver que o FlexUnit utiliza o runner do 
FlexUnit 4:

private function onCreationComplete():void {
 testRunner.runWithFlexUnit4Runner(currentRunTestSuite(), "FlexUnitExample");
}

No entanto, o framework é flexível, permitindo que você crie e utilize seus próprios 
runners, continuando com a mesma UI. De fato, atualmente existem runners para o 
FlexUnit 1, FlexUnit 4, Fluint e SLT.

O test runner no FlexUnit é um componente da UI que criará uma instância da suíte 
de testes e permitirá que você adicione todos os testes que deseja executar. O test 
runner também exibirá as informações do teste em seu navegador.

Escreva um caso de teste falho

Utilizando TDD, o próximo passo é escrever um teste falho. Diferentemente da pro-
gramação tradicional, você escreve o teste antes de escrever o código. Fazemos isso 
pensando antecipadamente nos nomes de classe que utilizaremos, nos métodos e no 
que os métodos terão de fazer. Neste caso, digamos que você tenha uma lista de em-
pregados, e deseja adicionar um empregado a ela. O teste será a criação da instância 
da classe utilitária que você criará futuramente: GetEmployeesInfo. Além disso, você está 
utilizando [Before] e [After] para definir a instância da classe, além de defini-la como 
null, quando concluída, como uma forma de evitar memory leaks.

package flexUnitTests {
 import org.flexunit.asserts.assertEquals;

 import utils.GetEmployeesInfo;

 public class GetEmployeesInfoTester {
  // declaração de referência para a classe a ser testada
  public var classToTestRef : utils.GetEmployeesInfo;

  [Before]

  public function setUpBeforeClass():void {
   classToTestRef = new GetEmployeesInfo();
  }
  [After]
  public function tearDownAfterClass():void {
   classToTestRef = null;
  }
  [Test]
  public function testAddItem():void {
   classToTestRef.addItem("John Do", "212-222-2222", "25", "john.do@gmail.com");
   assertEquals( classToTestRef.employeesCollection.getItemAt(0).name, "John Do" );
  }
 }
}

Programação orientada a objetos tem como base o princípio de que cada método 
tem um propósito, e a meta, portanto, é criar um teste que expresse o propósito desse 
método.

O stub testAddItem utiliza a classe de utilitário para adicionar um item, e verifica que 
o primeiro item da coleção foi adicionado corretamente.

Quando você compilar a aplicação, receberá os erros de tempo de compilação (Figura 
1.15), o que é produtivo, pois, dessa forma, o compilador diz a você o que deve ser 
feito a seguir.

Figura 1.15 – Testes no FlexUnit que mostram erros de tempo de compilação.

Escreva o código

Você encontrou as seguintes mensagens de erro no compilador, e elas devem ser 
solucionadas para que a aplicação rode:

•	 tipo	não	foi	encontrado	GetEmployeesInfo;

•	 chamada	a	um	método	possivelmente	indefinido	GetEmployeesInfo.

O primeiro código se deve à classe de utilitário faltante. Crie a classe e o pacote 
GetEmployeesInfo selecionando File ➤ New Package, e defina o nome como utils (Figura 1.16).

Figura 1.16 – Assistente de criação de novo pacote.

Em seguida, crie a classe GetEmployeesInfo. Selecione File ➤ New ActionScript class. Defina o nome 
como GetEmployeesInfo, e a superclasse para estender a classe flash.events.EventDispatcher. 
Selecione Finish (Figura 1.17).

package utils {
 import flash.events.EventDispatcher;
 import flash.events.IEventDispatcher;
 import mx.collections.ArrayCollection;
 import mx.rpc.events.FaultEvent;
 import mx.rpc.events.ResultEvent;
 import mx.rpc.http.HTTPService;

  public class GetEmployeesInfo extends EventDispatcher {
   private var service :HTTPService;
   private var _employeesCollection:ArrayCollection;

  public function GetEmployeesInfo() {
   _employeesCollection = new ArrayCollection();
  }



53Capítulo 1 ■ Criação de aplicações com desenvolvimento dirigido por testes

  public function get employeesCollection():ArrayCollection {
   return _employeesCollection;
  }

  public function addItem(name:String, phone:String, age:String, email:String):void {
   var item:Object = {name: name, phone: phone, age: age, email: email};
   employeesCollection.addItem( item );
  }
 }
}

Figura 1.17 – Assistente de criação de nova classe ActionScript.

O método testAddItem, essencialmente, toma um objeto e o adiciona a uma coleção. 
Assim, você pode facilmente testar essa ação, chamando o método e passando a 
informação de um empregado, para verificar que a coleção tem o mesmo valor que 
você adicionou.

public function testAddItem():void {
 classToTestRef.addItem("John Do", "212-222-2222", "25", "john.do@gmail.com");
 assertEquals( classToTestRef.employeesCollection.getItemAt(0).name, "John Do" );
}

Perceba também que você tem métodos de setUp e tearDown, que apenas criam uma 
nova instância da classe que estamos testando. Você está os definindo como null, 
quando termina.

Até aqui, no método testAddItem, você utilizou a asserção assertEquals, um método na 
classe Asserts ao qual tem acesso automático, já que está estendendo TestCase. A classe 
Asserts inclui muitas outras afirmações que você pode utilizar para testar seus métodos.

Uma vez compilada a aplicação, o erro de tempo de compilação desaparece.

Passe o teste

Para rodar os testes do FlexUnit, o Flash Builder 4 e o SDK adicionaram FlexUnit 
ao menu sob o ícone de “launch and debug” (Figura 1.18). Selecione o ícone de lan-
çamento ➤ Execute FlexUnit Tests.

Figura 1.18 – Execute o plug-in FlexUnit Tests.

Na próxima janela, você pode selecionar a suíte e os casos, ou o Test Case que gostaria 
de rodar. Neste caso, há apenas um método para ser testado (Figura 1.19).

Recomendamos a seleção da Test Suite ou dos Test Cases, mas não de ambos, 
já que isso rodará o teste duas vezes.

Figura 1.19 – Janela de configuração Run FlexUnit Tests.

Concluída a compilação, o navegador abre e mostra os resultados (Figura 1.20).

•	 1	teste	total	foi	executado;

•	 1	teve	sucesso;

•	 0	fracassou;

•	 0	apresentou	erro;

•	 0	foi	ignorado.

Figura 1.20 – Execução teste do FlexUnit no navegador.

Quando fechar o navegador, você poderá ver os resultados em FlexUnit Results (Figura 
1.21). Nessa tela, é possível ver que o teste passou, e que você recebeu uma luz verde.

Figura 1.21 – FlexUnit Results, visualização.

Uma vez escrito todo o código, e passado o teste, você sabe que seu teste atende a 
todos os requisitos de negócio. Por isso, pode compartilhar seu trabalho com o cliente, 
e com outros membros de sua equipe.

Refatore o código

Agora que seu teste passou, você pode refatorar o código, para deixá-lo pronto para 
produção. Por exemplo, pode adicionar um padrão de projeto para substituir um 
bloco de instruções if...else. Em nosso exemplo, não há nada a refatorar, pois o código 
é realmente simples.

Repita o processo, se necessário

Você pode continuar criando testes de unidade para a chamada de serviço, se quiser, 
por exemplo, recuperar um XML com informações de dados de empregados. Depois, 
você processa essas informações, adiciona todas a uma lista e despacha um evento, 
quando tiver concluído.

Escreva um teste falho para recuperar informações de empregados

Agora, escreva um teste para a chamada de serviço que recupere informações de 
empregados. Novamente, sua primeira tarefa será a criação do teste. Neste caso, você 
pode criar um evento personalizado, para passar as informações do empregado. Veja 
o método de teste:

[Test]
public function testLoad():void {
 classToTestRef.addEventListener( RetrieveInformationEvent.RETRIVE_INFORMATION,É
  addAsync( onResult, 500 ));
 classToTestRef.load("assets/file.xml");
}

[Test]
public function onResult(event:RetrieveInformationEvent):void {
  assertNotNull( event.employeesCollection );
}

O método testLoad adiciona um ouvinte de evento, para que você possa esperar a 
conclusão da chamada assíncrona, manipulando os resultados no método onResult. 
Perceba que você está utilizando a função addAsync, o que indica que aguardará 500 
ms para a conclusão da chamada.

O método onResult confere, para garantir que você obteve os resultados. Não importa 
o tipo dos resultados deste teste. Você está apenas verificando que o método load 
recebeu os resultados e os adicionou à coleção. É possível criar outro teste, para ve-
rificar a integridade dos dados.

Escreva código para recuperar informações de empregados

Veja a classe GetEmployeesInfoTester.as completa.

package flexUnitTests {
 import org.flexunit.asserts.assertEquals;
 import org.flexunit.asserts.assertNotNull;
 import org.flexunit.async.Async;

 import utils.GetEmployeesInfo;
 import utils.events.RetrieveInformationEvent;

 public class GetEmployeesInfoTester {
  // declaração de referência para a classe a ser testada
  public var classToTestRef : GetEmployeesInfo;

  [Before]
  public function setUpBeforeClass():void {
   classToTestRef = new GetEmployeesInfo();
  }

  [After]
  public function tearDownAfterClass():void {
   classToTestRef = null;
  }

  [Test]
  public function testAddItem():void {
   classToTestRef.addItem("John Do", "212-222-2222", "25", "john.do@gmail.com");
   assertEquals( classToTestRef.employeesCollection.getItemAt(0).name, "John Do" );
  }

  [Test(async,timeout="500")]
  public function testLoad():void {
   classToTestRef.addEventListener( RetrieveInformationEventÉ
   .RETRIEVE_INFORMATION, Async.asyncHandler( this,  onResult, 500 ), false, 0, true );
   classToTestRef.load("assets/file.xml");
  }

  public function onResult(event:RetrieveInformationEvent, passThroughData:Object):void {
   assertNotNull( event.employeesCollection );
  }
 }
}

Compile a classe, e observe a presença de erros de tempo de compilação. Uma vez mais, 
esses erros indicam o que você deve fazer a seguir. Retorne à classe GetEmployeesInfo.
as, e adicione um método load, para carregar o XML. Veja o código completo a seguir:

package utils {
 import flash.events.EventDispatcher;
 import flash.events.IEventDispatcher;
 import mx.collections.ArrayCollection;



Flex 4 Avançado58

 import mx.rpc.events.FaultEvent;
 import mx.rpc.events.ResultEvent;
 import mx.rpc.http.HTTPService;
 import utils.events.RetrieveInformationEvent;

 public class GetEmployeesInfo extends EventDispatcher {
  private var service :HTTPService;
  private var _employeesCollection:ArrayCollection;

  public function GetEmployeesInfo() {
   _employeesCollection = new ArrayCollection();
  }

  public function get employeesCollection():ArrayCollection {
   return _employeesCollection;
  }

  public function load(file:String):void  {
   service = new HTTPService();
   service.url = file;
   service.resultFormat = "e4x";
   service.addEventListener(ResultEvent.RESULT, onResult);
   service.addEventListener(FaultEvent.FAULT, onFault);

   service.send();
  }

  private function onResult(event:ResultEvent):void {
   var employees:XML = new XML( event.result );
   var employee:XML;

   for each (employee in employees.employee) {
    this.addItem( employee.name, employee.phone, employee.age, employee.email );
   }

   this.dispatchEvent( new RetrieveInformationEvent( employeesCollection ));
  }

  private function onFault(event:FaultEvent):void {
   trace("errors loading file");
  }

  public function addItem(name:String, phone:String, age:String, email:String):void {
   var item:Object = {name: name, phone: phone, age: age, email: email};
   employeesCollection.addItem( item );
  }
 }
}

Você tem uma variável para criar uma instância da classe de serviço HTTPService, que 
será utilizada para executar a chamada de serviço.

private var service:HTTPService;

A coleção é a variável que você utilizará para armazenar os resultados. Você não quer 
que uma subclasse tenha a habilidade de alterar a coleção diretamente, mas quer 
permitir o acesso à coleção, para que possamos posicionar um setter.

private var _employeesCollection:ArrayCollection;

public function GetEmployeesInfo() {
 _employeesCollection = new ArrayCollection();
}

O método GetEmployeesInfo é seu construtor default. Apenas crie um novo objeto para 
a ArrayCollection.

public function GetEmployeesInfo() {
 employeesCollection = new ArrayCollection();
}

O método load será utilizado para apontar para o arquivo que você carregará. Adicione 
ouvintes e inicie o serviço.

public function load(file:String):void {
 service = new HTTPService();
 service.url = file;
 service.resultFormat = "e4x";
 service.addEventListener(ResultEvent.RESULT, onResult);
 service.addEventListener(FaultEvent.FAULT, onFault);

 service.send();
}

O método onResult será invocado em uma chamada de serviço de sucesso. Você itera 
pela classe, e toma os dados, utilizando o método addItem para adicionar o empregado 
à coleção. Concluído o processamento, você utiliza o método dispatchEvent para des-
pachar um evento RetrieveInformationEvent e passar a employeesCollection.

 private function onResult(event:ResultEvent):void {
  var employees:XML = new XML( event.result );
  var employee:XML;

  for each (employee in employees.employee) {
   this.addItem( employee.name, employee.phone, employee.age, employee.email );
  }

  this.dispatchEvent( new RetrieveInformationEvent( employeesCollection ));
 }

O método onFault será chamado nos casos em que você encontrar erros, enquanto 
tenta efetuar a chamada de serviço, como ao chamar um arquivo que não existe, ou 
na presença de erros de segurança. Neste ponto, você está deixando esse método 
com um segmento trace, porém se sinta livre para implementar o despacho de um 
evento de erro.

private function onFault(event:FaultEvent):void {
 trace("errors loading file");
}

O método addItem será utilizado para pegar os parâmetros que você está passando, 
adicionando-os como objetos à coleção.

  public function addItem(name:String, phone:String, age:String, email:String):void {
   var item:Object = {name: name, phone: phone, age: age, email: email};
   employeesCollection.addItem( item );
  }
 }
}

A variável getEmployeesInfo armazenará a instância da classe de utilitário.

private var getEmployeesInfo:GetEmployeesInfo;

Assim que a aplicação despachar o evento applicationComplete, applicationCompleteHandler 
será chamado, e criará uma nova instância da classe GetEmployeesInfo, utilizando o 
método load, para iniciar o serviço e passar a localização do XML.

protected function applicationCompleteHandler(event:FlexEvent):void {
 getEmployeesInfo = new GetEmployeesInfo();
 getEmployeesInfo.load("assets/file.xml");

 getEmployeesInfo.addEventListener(RetrieveInformationEvent.RETRIVE_INFORMATION, 
onInfoRetrieved );
}

O XML que utilizaremos é simples. Crie um arquivo, assets/file.xml, e adicione o 
seguinte código: 

<employees>
 <employee>
  <name>John Do</name>
  <phone>212-222-2222</phone>
  <age>20</age>
  <email>john@youremail.com</email>
 </employee>

 <employee>
  <name>Jane Smith</name>
  <phone>212-333-3333</phone>
  <age>21</age>
  <email>jane@youremail.com</email>
 </employee>
</employees>

Por fim, uma vez despachado o evento RetrieveInformationEvent, manipularemos o evento, 
mostrando o atributo de nome do primeiro item.

private function onInfoRetrieved(event:RetrieveInformationEvent):void {
 trace(event.employeesCollection.getItemAt(0).name);
}

Crie um evento personalizado, RetrieveInformationEvent, que armazena, e a coleção array 
employeesCollection:

package utils.events {
 import flash.events.Event;
 import mx.collections.ArrayCollection;
 public class RetrieveInformationEvent extends Event {
  public static const RETRIVE_INFORMATION:String = "RetrieveInformationEvent";
  public var employeesCollection:ArrayCollection;
  public function RetrieveInformationEvent( employeesCollection:ArrayCollection ) {
   this.employeesCollection = employeesCollection;
   super(RETRIVE_INFORMATION, false, false);
  }
 }
}

Compile e rode a aplicação. Você pode ver os resultados da instrução trace na janela 
do console.

Passe o teste

Agora, tornamos sua aplicação capaz de efetuar uma chamada de serviço, e de re-
cuperar informações de empregados. Sem modificar GetEmployeesInfoTester, você pode 
rodar o test runner, e deve receber uma luz verde (Figura 1.22).



Flex 4 Avançado62

Figura 1.22 – Executando testes no FlexUnit.

Refatore

A única refatoração que nos interessa é a adição de uma tag de metadados que 
aponte para o evento RetrieveInformationEvent, para que você tenha dicas de código 
quando implementar a classe no Flex. Adicione o código seguinte ao início da classe 
GetEmployeesInfo:

 [Event(name="retriveInformation", type="utils.events.RetrieveInformationEvent")]

Você pode baixar o exemplo completo em http://www.friendsofed.com/download.
html?isbn=1430224835.

Resumo
Neste capítulo, abordamos o FlexUnit 4 e o desenvolvimento dirigido por testes (TDD). 
Partimos de uma visão geral do FlexUnit 4, e, depois, mostramos como criar as classes 
de suíte de teste, os caso de teste e o test runner. Abordamos todos os métodos de 
asserção disponíveis no FlexUnit 4, os testes assíncronos e a manipulação de exceções. 
Também tratamos de métodos de asserção adicionais do FlexUnit 4 (como os ofere-
cidos pela biblioteca Hamcrest), de teorias, testes e dos testes de interfaces de usuário.

Na segunda parte do capítulo, discutimos o desenvolvimento dirigido por testes com 
o FlexUnit 4. Mostramos como criar um caso de teste falho, além de como escrever o 
código, passar um teste e refatorar esse código. Esperamos ter oferecido a inspiração 
necessária para que você utilize a técnica TDD em suas aplicações Flash móveis, web 
e desktop, escrevendo um código melhor, mais escalável e reutilizável.


